USGS
science for a changing world

Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

A MODULAR THREE-DIMENSIONAL FINITE-DIFFERENCE GROUND-WATER FLOW MODEL

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological Survey Open-File Report 83–875

Book 6

MODELING TECHNIQUES

Narrative for Module RCH1RP

This module reads data used to calculate the terms which represent areally distributed recharge.

1. Read the values INRECH and INIRCH which indicate whether the data contained in arrays RECH and IRCH used during the last stress period are to be used for the current stress period.

2. Test INRECH to see where the recharge flux (RECH) is coming from. If INRECH is less than zero, the recharge rate used in the last stress period will be used again in this stress period. Print a message to that effect. GO TO STEP 5.

3. If INRECH is greater than or equal to zero, CALL U2DREL to read the recharge rate (RECH).

4. Multiply the specified recharge flux rates by the cell areas to get the volumetric-recharge rate.

5. If the recharge option (NRCHOP) is not equal to two, a layer-indicator array is not needed. GO TO STEP 8.

6. If INIRCH is less than zero, the data in IRCH left over from the last stress period will be used in this stress period. Print a message to that effect. GO TO STEP 8.

7. If INIRCH is greater than or equal to zero, CALL U2DINT to read the IRCH array.

8. RETURN.

Flow Chart for Module RCH1RP

INRECH is a flag which, when set, indicates that recharge rates (RECH) should be read for the current stress period. If it is clear (< 0), recharge rates from the last stress period should be reused.

INIRCH is a flag similar to INRECH used for the layer indicator array IRCH.

RECH is an array containing a recharge rate for every horizontal cell location.

IRCH is an array containing a recharge indicator for each horizontal cell location. For each horizontal cell location, it indicates the layer number of the cell at that location which gets recharge. It is used only if the recharge option (NRCHOP) is equal to two.



```
      SUBROUTINE RCH1RP(NRCHOP,IRCH,RECH,DELR,DELC,NROW,NCOL,
     1                  IN,IOUT)
C
C-----VERSION 1634 24JUL1987 RCH1RP
C     ******************************************************************
C     READ RECHARGE RATES
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 ANAME
      DIMENSION IRCH(NCOL,NROW),RECH(NCOL,NROW),
     1          ANAME(6,2),DELR(NCOL),DELC(NROW)
C
      DATA ANAME(1,1),ANAME(2,1),ANAME(3,1),ANAME(4,1),ANAME(5,1),
     1  ANAME(6,1) /'    ','RECH','ARGE',' LAY','ER I','NDEX'/
      DATA ANAME(1,2),ANAME(2,2),ANAME(3,2),ANAME(4,2),ANAME(5,2),
     1  ANAME(6,2) /'    ','    ','    ','    ','RECH','ARGE'/
C     ------------------------------------------------------------------
C
C1------READ FLAGS SHOWING WHETHER DATA IS TO BE REUSED.
      READ(IN,4)INRECH,INIRCH
    4 FORMAT(2I10)
C
C2------TEST INRECH TO SEE WHERE RECH IS COMING FROM.
      IF(INRECH.GE.0)GO TO 32
C
C2A-----IF INRECH<0 THEN REUSE RECHARGE ARRAY FROM LAST STRESS PERIOD
      WRITE(IOUT,3)
    3 FORMAT(1HO,'REUSING RECH FROM LAST STRESS PERIOD')
      GO TO 55
C
C3------IF INRECH=>0 THEN CALL U2DREL TO READ RECHARGE RATE.
   32 CALL U2DREL(RECH,ANAME(1,2),NROW,NCOL,0,IN,IOUT)
C
C4------MULTIPLY RECHARGE RATE BY CELL AREA TO GET VOLUMETRIC RATE.
      DO 50 IR=1,NROW
      DO 50 IC=1,NCOL
      RECH(IC,IR)=RECH(IC,IR)*DELR(IC)*DELC(IR)
   50 CONTINUE
C
C5------IF NRCHOP=2 THEN A LAYER INDICATOR ARRAY IS NEEDED.
  55   IF (NRCHOP.NE.2)GO TO 60
C
C6------IF INIRCH<0 THEN REUSE LAYER INDICATOR ARRAY.
      IF(INIRCH.GE.0)GO TO 58
      WRITE(IOUT,2)
    2 FORMAT(1HO,'REUSING IRCH FROM LAST STRESS PERIOD')
      GO TO 60
C
C7------IF INIRCH=>0 CALL U2DINT TO READ LAYER IND ARRAY(IRCH)
   58 CALL U2DINT(IRCH,ANAME(1,1),NROW,NCOL,0,IN,IOUT)
C
C8------RETURN
   60 RETURN
      END
```

List of Variables for Module RCH1RP

Variable	Range	Definition
ANAME	Module	Label for printout of the input array.
DELC	Global	DIMENSION (NROW), Cell dimension in the column direction. DELC(I) contains the width of row I.
DELR	Global	DIMENSION (NCOL), Cell dimension in the row direction. DELR(J) contains the width of column J.
IC	Module	Index for columns.
IN	Package	Primary unit number from which input for this package will be read.
INIRCH	Module	Flag. $\geq$ 0, IRCH array will be read. < 0, IRCH array already in memory from the last stress period will be used.
INRECH	Module	Flag. $\geq$ 0, RECH array will be read. < 0, RECH array already in memory from the last stress period will be used.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
IRCH	Package	DIMENSION (NCOL,NROW), Layer number for each horizontal cell location to which recharge will be applied if the recharge option (NRCHOP) is equal to 2.
NCOL	Global	Number of columns in the grid.
NRCHOP	Package	Recharge option: = 1, recharge is to the top grid layer. = 2, recharge is to the grid layer specified in array IRCH. = 3, recharge is to the highest variable-head cell which is not below a constant-head cell.
NROW	Global	Number of rows in the grid.
RECH	Package	DIMENSION (NCOL,NROW), Recharge flow rate. Recharge flux is read into RECH and than multiplied by cell area to obtain recharge flow rate.

Narrative for Module RCH1FM

This module adds terms representing areally distributed recharge to the accumulators in which the terms HCOF and RHS are formulated.

1. If the recharge option (NRCHOP) is equal to one, recharge is to the top layer. For each horizontal location, DO STEPS (a) AND (b).

   (a) If the cell is external (IBOUND(I,J,K) $\leq$ 0), ignore it. SKIP STEP (b).

   (b) Subtract the recharge flow rate from the RHS accumulator.

2. If the recharge option is two, recharge is only to the cells specified in the layer-indicator array (IRCH).

   (a) Get the layer index from the layer-indicator array (IRCH).

   (b) If the cell is external, ignore it. SKIP STEP (c).

   (c) Subtract the recharge flow rate from the RHS accumulator.

3. If the recharge option is three, recharge is in the uppermost internal cell. For each horizontal cell location:

   (a) If the cell is constant head, there will be no recharge below it. Move on to the next horizontal cell location.

   (b) If the cell is no flow, move down a cell and go back to (a).

   (c) Subtract the recharge flow rate from the RHS accumulator. Move on to the next horizontal cell location.

4. RETURN

Flow Chart for Module RCH1FM

RHS is the right hand side of the finite-difference equation. It includes all terms that are independent of head at the end of the time step.

IRCH is an array which contains the layer number to which recharge is applied for each horizontal location. It is used only if option 2 has been specified.

NRCHOP is the recharge option.

1 - Recharge is to the top layer.

2 - Recharge is to the layer specified by the user in the indicator array (IRCH).

3 - Recharge is to the uppermost active cell.



```
      SUBROUTINE RCH1FM(NRCHOP,IRCH,RECH,RHS,IBOUND,NCOL,
     1                  NROW,NLAY)
C
C-----VERSION 1404 12MAY1987 RCH1FM
C     ******************************************************************
C     SUBTRACT RECHARGE FROM RHS
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      DIMENSION IRCH(NCOL,NROW),RECH(NCOL,NROW),
     1          RHS(NCOL,NROW,NLAY),IBOUND(NCOL,NROW,NLAY)
C     ------------------------------------------------------------------
C
C1------IF NRCHOP IS 1 RECHARGE IS IN TOP LAYER. LAYER INDEX IS 1.
      IF(NRCHOP.NE.1) GO TO 15
C
      DO 10 IR=1,NROW
      DO 10 IC=1,NCOL
C
C1A-----IF CELL IS EXTERNAL THERE IS NO RECHARGE INTO IT.
      IF(IBOUND(IC,IR,1).LE.0)GO TO 10
C
C1B-----SUBTRACT RECHARGE RATE FROM RIGHT-HAND-SIDE.
      RHS(IC,IR,1)=RHS(IC,IR,1)-RECH(IC,IR)
   10 CONTINUE
      GO TO 100
C
C2------IF OPTION IS 2 THEN RECHARGE IS INTO LAYER IN INDICATOR ARRAY
   15 IF(NRCHOP.NE.2)GO TO 25
      DO 20 IR=1,NROW
      DO 20 IC=1,NCOL
C
C2A-----LAYER INDEX IS IN INDICATOR ARRAY.
      IL=IRCH(IC,IR)
C
C2B-----IF THE CELL IS EXTERNAL THERE IS NO RECHARGE INTO IT.
      IF(IBOUND(IC,IR,IL).LE.0)GO TO 20
C
C2C-----SUBTRACT RECHARGE FROM RIGHT-HAND-SIDE.
      RHS(IC,IR,IL)=RHS(IC,IR,IL)-RECH(IC,IR)
   20 CONTINUE
      GO TO 100
C
C3------IF OPTION IS 3 RECHARGE IS INTO HIGHEST INTERNAL CELL.
   25 IF(NRCHOP.NE.3)GO TO 100
C        CANNOT PASS THROUGH CONSTANT HEAD NODE
      DO 30 IR=1,NROW
      DO 30 IC=1,NCOL
      DO 28 IL=1,NLAY
C
C3A-----IF CELL IS CONSTANT HEAD MOVE ON TO NEXT HORIZONTAL LOCATION.
      IF(IBOUND(IC,IR,IL).LT.0) GO TO 30
C
C3B-----IF CELL IS INACTIVE MOVE DOWN A LAYER.
      IF (IBOUND(IC,IR,IL).EQ.0)GO TO 28
C
C3C-----SUBTRACT RECHARGE FROM RIGHT-HAND-SIDE.
      RHS(IC,IR,IL)=RHS(IC,IR,IL)-RECH(IC,IR)
      GO TO 30
   28 CONTINUE
   30 CONTINUE
  100 CONTINUE
C
C4------RETURN
      RETURN
      END
```

List of Variables for Module RCH1FM

Variable	Range	Definition
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
IRCH	Package	DIMENSION (NCOL,NROW), Layer number for each horizontal cell location to which recharge will be applied if the recharge option (NRCHOP) is equal to 2.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NRCHOP	Package	Recharge option: = 1, recharge is to the top grid layer. = 2, recharge is to the grid layer specified in array IRCH. = 3, recharge is to the highest variable-head cell which is not below a constant-head cell.
NROW	Global	Number of rows in the grid.
RECH	Package	DIMENSION (NCOL,NROW), Recharge flow rate.
RHS	Global	DIMENSION (NCOL,NROW,NLAY), Right hand side of the finite-difference equation. RHS is an accumulation of terms from several different packages.

Narrative for Module RCH1BD

This module calculates rates and volumes added to the aquifer by areally distributed recharge.

1. Clear the rate accumulators RATIN and RATOUT.

2. If cell-by-cell flow terms will be saved, clear the buffer (BUFF) in which they will be accumulated.

3. If the recharge option is one, the recharge goes into the top layer. Process the horizontal locations one at a time.

   (a) If the cell is external, do not calculate budget.

   (b) If cell-by-cell flow terms will be saved, add recharge to the buffer.

   (c) If the recharge is positive, add it to RATIN; otherwise, add it to RATOUT.

4. If the recharge option is two, recharge goes into the layer specified in indicator array (IRCH). Process the horizontal locations one at a time.

   (a) Get the cell layer from indicator array (IRCH).

   (b) If the cell is external, do not calculate budget.

   (c) If cell-by-cell flow terms will be saved, add the recharge to the buffer.

   (d) If the recharge is positive, add it to RATIN; otherwise, add it to RATOUT.

5. If the recharge option is three, the recharge goes into the top variable-head cell provided there is not a constant-head cell above it. Process the horizontal locations one at a time. Start with the top cell and work down.

(a) If the cell is inactive, there is no recharge into that cell; move down to the next one.

(b) If the cell is constant, there is no recharge at this horizontal location; move on to the next horizontal location.

(c) If cell-by-cell flow terms are to be saved, add the recharge to the buffer.

(d) If the recharge is positive, add it to RATIN; otherwise, add it to RATOUT.

6. If cell-by-cell flow terms will be saved, call module UBUDSV to write the buffer (BUFF) onto disk.

7. Move RATIN and RATOUT into the VBVL array for printing by BAS1OT.

8. Add RATOUT multiplied by the time-step length to the volume accumulators in VBVL for printing by BAS1OT.

9. Move the recharge budget-term labels to VBNM for printing by BAS1OT.

10. Increment the budget-term counter (MSUM).

11. RETURN.

Flow Chart for Module RCH1BD

RATIN is an accumulator to which all flows into the aquifer are added.

RATOUT is an accumulator to which all flows out of the aquifer are added.

BUFFER is an array in which values are stored as they are being gathered for printing or recording.

NRCHOP is the recharge option.

1 - Recharge is to the top layer.

2 - Recharge is to the layer specified by the user in the indicator array (IRCH).

3 - Recharge is to the uppermost active cell.

IRCH is an array containing a recharge indicator for each horizontal cell. It is used only if the recharge option (NRCHOP) is equal to two.

VBVL is a table of budget entries calculated by component-of-flow packages for use in calculating the volumetric budget.

VBNM is a table of labels for budget terms.

EXTERNAL: a cell is external if it is either no flow (inactive) or constant head.



```
      SUBROUTINE RCH1BD(NRCHOP,IRCH,RECH,IBOUND,NROW,NCOL,NLAY,
     1    DELT,VBVL,VBNM,MSUM,KSTP,KPER,IRCHCB,ICBCFL,BUFF,IOUT)
C
C-----VERSION 1602 12MAY1987 RCH1BD
C     ******************************************************************
C     CALCULATE VOLUMETRIC BUDGET FOR RECHARGE
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 VBNM,TEXT
      DIMENSION IRCH(NCOL,NROW),RECH(NCOL,NROW),
     1          IBOUND(NCOL,NROW,NLAY),BUFF(NCOL,NROW,NLAY),
     2          VBVL(4,20),VBNM(4,20)
      DIMENSION TEXT(4)
      DATA TEXT(1),TEXT(2),TEXT(3),TEXT(4) /'    ','    ','RECH','ARGE'/
C     ------------------------------------------------------------------
C
C1------CLEAR THE RATE ACCUMULATORS.
      RATIN=0.
      RATOUT=0.
C
C2------IF CELL-BY-CELL FLOW TERMS WILL BE SAVED THEN CLEAR THE BUFFER.
      IBD=0
      IF(ICBCFL.EQ.0 .OR. IRCHCB.LE.0) GO TO 5
      IBD=1
      DO 2 IL=1,NLAY
      DO 2 IR=1,NROW
      DO 2 IC=1,NCOL
      BUFF(IC,IR,IL)=0.
    2 CONTINUE
C
C3------IF NRCHOP=1 RECH GOES INTO LAYER 1. PROCESS EACH HORIZONTAL
C3------CELL LOCATION.
    5 IF(NRCHOP.NE.1) GO TO 15
C
C     ---RECHARGE IS APPLIED TO TOP LAYER
      DO 10 IR=1,NROW
      DO 10 IC=1,NCOL
C
C3A-----IF CELL IS EXTERNAL THEN DO NOT DO BUDGET FOR IT.
      IF(IBOUND(IC,IR,1).LE.0)GO TO 10
      Q=RECH(IC,IR)
C
C3B-----IF CELL-BY-CELL FLOW TERMS WILL BE SAVED THEN ADD RECH TO BUFF
      IF(IBD.EQ.1) BUFF(IC,IR,1)=Q
C
C3C-----IF RECH POSITIVE ADD IT TO RATIN ELSE ADD IT TO RATOUT.
      IF(Q) 8,10,7
    7 RATIN=RATIN+Q
      GO TO 10
    8 RATOUT=RATOUT-Q
   10 CONTINUE
      GO TO 100
C
C4------IF NRCHOP=2 RECH IS IN LAYER SHOWN IN INDICATOR ARRAY(IRCH).
C4------PROCESS HORIZONTAL CELL LOCATIONS ONE AT A TIME.
   15 IF(NRCHOP.NE.2)GO TO 25
      DO 20 IR=1,NROW
      DO 20 IC=1,NCOL
C
C4A-----GET LAYER INDEX FROM INDICATOR ARRAY(IRCH).
      IL=IRCH(IC,IR)
C
C4B-----IF CELL IS EXTERNAL DO NOT CALCULATE BUDGET FOR IT.
      IF(IBOUND(IC,IR,IL).LE.0)GO TO 20
```

```
      Q=RECH(IC,IR)
C
C4C-----IF C-B-C FLOW TERMS WILL BE SAVED THEN ADD RECHARGE TO BUFFER.
      IF(IBD.EQ.1) BUFF(IC,IR,IL)=Q
C
C4D-----IF RECHARGE IS POSITIVE ADD TO RATIN ELSE ADD IT TO RATOUT.
      IF(Q) 18,20,17
   17 RATIN=RATIN+Q
      GO TO 20
   18 RATOUT=RATOUT-Q
   20 CONTINUE
      GO TO 100
C
C5------IF OPTION=3 RECHARGE IS INTO HIGHEST INTERNAL CELL. IT WILL NOT
C5------PASS THROUGH A CONSTANT HEAD CELL. PROCESS HORIZONTAL CELL
C5------LOCATIONS ONE AT A TIME.
   25 IF(NRCHOP.NE.3)GO TO 100
      DO 30 IR=1,NROW
      DO 30 IC=1,NCOL
      DO 28 IL=1,NLAY
C
C5A-----IF CELL IS CONSTANT HEAD MOVE ON TO NEXT HORIZONTAL LOCATION.
      IF(IBOUND(IC,IR,IL).LT.0) GO TO 30
C
C5B-----IF CELL IS INACTIVE MOVE DOWN TO NEXT CELL.
      IF (IBOUND(IC,IR,IL).EQ.0)GO TO 28
      Q=RECH(IC,IR)
C
C5C-----IF C-B-C FLOW TERMS TO BE SAVED THEN ADD RECHARGE TO BUFFER.
      IF(IBD.EQ.1) BUFF(IC,IR,IL)=Q
C
C5D-----IF RECH IS POSITIVE ADD IT TO RATIN ELSE ADD IT TO RATOUT.
      IF(Q) 27,30,26
   26 RATIN=RATIN+Q
      GO TO 30
   27 RATOUT=RATOUT-Q
      GO TO 30
   28 CONTINUE
   30 CONTINUE
C
  100 CONTINUE
C
C6------IF C-B-C FLOW TERMS TO BE SAVED CALL UBUDSV TO WRITE THEM.
      IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,IRCHCB,BUFF,NCOL,NROW,
     1                         NLAY,IOUT)
C
C7------MOVE TOTAL RECHARGE RATE INTO VBVL FOR PRINTING BY BAS1OT.
      VBVL(4,MSUM)=RATOUT
      VBVL(3,MSUM)=RATIN
C
C8------ADD RECHARGE FOR TIME STEP TO RECHARGE ACCUMULATOR IN VBVL.
      VBVL(2,MSUM)=VBVL(2,MSUM)+RATOUT*DELT
      VBVL(1,MSUM)=VBVL(1,MSUM)+RATIN*DELT
C
C9------MOVE BUDGET TERM LABELS TO VBNM FOR PRINT BY MODULE BAS_OT.
      VBNM(1,MSUM)=TEXT(1)
      VBNM(2,MSUM)=TEXT(2)
      VBNM(3,MSUM)=TEXT(3)
      VBNM(4,MSUM)=TEXT(4)
C
C10-----INCREMENT BUDGET TERM COUNTER.
      MSUM=MSUM+1
C
C11-----RETURN
      RETURN
      END
```

List of Variables for Module RCH1BD

Variable	Range	Definition
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
DELT	Global	Length of the current time step.
IBD	Module	Flag. = 0, cell-by-cell flow terms for this package will not be recorded. ≠ 0, cell-by-cell flow terms for this package will be recorded.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
ICBCFL	Global	Flag. = 0, cell-by-cell flow terms will not be recorded or printed for the current time step. ≠ 0, cell-by-cell flow terms will be recorded for the current time step.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
IRCH	Package	DIMENSION (NCOL,NROW), Layer number for each horizontal cell location to which recharge will be applied if the recharge option (NRCHOP) is equal to 2.
IRCHCB	Package	Flag. IRCHCB ≤ 0, cell-by-cell flow terms will not be recorded or printed. IRCHCB > 0 and ICBCFL ≠ 0, cell-by-cell flow terms for the RCH1 Package will be recorded on UNIT = IRCHCB.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
NCOL	Global	Number of columns in the grid.

List of Variables for Module RCH1BD (Continued)

Variable	Range	Definition
NLAY	Global	Number of layers in the grid.
NRCHOP	Package	Recharge option: = 1, recharge is to the top grid layer. = 2, recharge is to the grid layer specified in array IRCH. = 3, recharge is to the highest variable-head cell which is not below a constant-head cell.
NROW	Global	Number of rows in the grid.
Q	Module	Flow from recharge into a cell. (Reverse the sign to get flow out of the cell.)
RATIN	Module	Accumulator for the total flow into the flow field from recharge.
RATOUT	Module	Accumulator for the total flow out of the flow field to recharge.
RECH	Package	DIMENSION (NCOL,NROW), Recharge flow rate.
TEXT	Module	Label to be printed or recorded with the array data.
VBNM	Global	DIMENSION (4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION (4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N), Rate for the current time step into the flow field. (2,N), Rate for the current time step out of the flow field. (3,N), Volume into the flow field during simulation. (4,N), Volume out of the flow field during simulation.

CHAPTER 8

WELL PACKAGE

Conceptualization and Implementation

The Well Package is designed to simulate features such as wells which withdraw water from the aquifer (or add water to it) at a specified rate during a given stress period, where the rate is independent of both the cell area and the head in the cell. The discussion in this section is developed on the assumption that the features to be simulated are actually wells, either discharging or recharging.

Well discharge is handled in the Well Package by specifying the rate, Q, at which each individual well adds water to the aquifer or removes water from it, during each stress period of the simulation. Negative values of Q are used to indicate well discharge, while positive values of Q indicate a recharging well.

At the beginning of each stress period, the WEL1RP module reads four values for each well--the row, column and layer number of the cell in which the well is located, and the discharge or recharge rate, Q, of the well during that stress period. At each iteration, as the matrix equations are formulated, the value of Q for each well is subtracted from the RHS value (equation (26) or (29)) for the cell containing that well. Where more than one well falls within a single cell, the calculation is repeated for each well as the RHS term for that cell is assembled. Thus the user specifies the discharge associated with each individual well, and these are in effect summed within the program to obtain the total discharge from the cell.

The Well Package, as it is presently formulated, does not accommodate wells which are open to more than one layer of the model. However, a well of this type can be represented as a group of single-layer wells, each open to one of the layers tapped by the multi-layer well, and each having an individual Q term specified for each stress period. If this approach is used, the discharge of the multi-layer well must be divided or apportioned in some way among the individual layers, externally to the model program. A common method of doing this is to divide the well discharge in proportion to the layer transmissivities i.e.

$$\frac{Q_l}{Q_w} = \frac{T_l}{\Sigma T} \tag{68}$$

where Q_l is the discharge from layer l to a particular well in a given stress period, Q_w is the well discharge in that stress period, T_l is the transmissivity of layer l and ΣT represents the sum of the transmissivities of all layers penetrated by the well. Again, it's important to note that equation (68), or some other method of apportioning the discharge, must be implemented by the user externally to the program for each multi-layer well, and for each stress period.

This approach, in which a multi-layer well is represented as a group of single layer wells, fails to take into account the interconnection between various layers provided by the well itself, and is thus an incomplete representation of the problem. A package which will provide an improved approximation of multi-layer well effects is under development.

Well Package Input

Input for the Well (WEL) Package is read from the unit specified in IUNIT(2).

FOR EACH SIMULATION

WEL1AL

1. Data: MXWELL IWELCB
   Format: I10 I10

FOR EACH STRESS PERIOD

WEL1RP

2. Data: ITMP
   Format: I10

3. Data: Layer Row Column Q
   Format: I10 I10 I10 F10.0

   (Input item 3 normally consists of one record for each well. If ITMP is negative or zero, item 3 is not read.)

Explanation of Fields Used in Input Instructions

MXWELL--is the maximum number of wells used at any time.

IWELCB--is a flag and a unit number.

If IWELCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see Output Control) is set.

If IWELCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IWELCB < 0, well recharge will be printed whenever ICBCFL is set.

ITMP--is a flag and a counter.

If ITMP < 0, well data from the last stress period will be reused.

If ITMP $\geq$ 0, ITMP will be the number of wells active during the current stress period.

Layer--is the layer number of the model cell that contains the well.

Row--is the row number of the model cell that contains the well.

Column--is the column number of the model cell that contains the well.

Q--is the volumetric recharge rate. A positive value indicates recharge and a negative value indicates discharge.

SAMPLE INPUT TO THE WELL PACKAGE

DATA ITEM	EXPLANATION	INPUT RECORDS			
1	{MXWELL, IWELCB}	6	24		
2	STRESS PERIOD 1 {ITMP}	4			
3	FIRST WELL {Layer, Row, Column, Q	2	5	6	-.17
3	SECOND WELL {Layer, Row, Column, Q}	2	6	3	-.23
3	THIRD WELL {Layer, Row, Column, Q}	2	7	7	-.77
3	FOURTH WELL {Layer, Row, Column, Q}	2	9	2	-.32
2	STRESS PERIOD 2 {ITMP}	-1			
2	STRESS PERIOD 3 {ITMP}	-1			
2	STRESS PERIOD 4 {ITMP}	6			
3	FIRST WELL {Layer, Row, Column, Q}	2	6	5	-.73
3	SECOND WELL {Layer, Row, Column, Q}	2	9	4	-.32
3	THIRD WELL {Layer, Row, Column, Q	2	5	6	-.17
3	FOURTH WELL {Layer, Row, Column, Q}	2	6	3	-.43
3	FIFTH WELL {Layer, Row, Column, Q}	2	7	7	-.77
3	SIXTH WELL {Layer, Row, Column, Q}	2	9	2	-.32

Module Documentation for the Well Package

The Well Package (WEL1) consists of four modules, all of which are called by the MAIN program. The modules are:

WEL1AL — Allocates space for the list of wells (WELL).

WEL1RP — Reads location and Q value (discharge or recharge rate) for all wells.

Note: Q is entered as a negative number for well discharge and as a positive number for well recharge.

WEL1FM — Subtracts Q values from the term RHS for each cell containing pumping wells.

WEL1BD — Calculates the rates and accumulated volume of recharge to or discharge from the flow system by pumping wells.

Narrative for Module WEL1AL

This module allocates space in the X array to store the list of wells. The X array is a pool of memory space from which space is allocated for tables, lists, and arrays.

1. Print a message identifying the package and initialize NWELLS (a counter containing the number of wells).

2. Read and print MXWELL (the maximum number of wells) and IWELBD (the unit number for cell-by-cell flow terms or a flag indicating that cell-by-cell flow terms should be printed).

3. Set LCWELL, which will point to the first element in the well list (WELL), equal to ISUM, which is currently pointing to the first unallocated element in the X array.

4. Calculate the amount of space needed for the well list (four values for each cell--row, column, layer, and rate) and add it to ISUM.

5. Print the number of elements in the X array used by the Well Package.

6. If the pointer to the lowest unallocated element in the X array (ISUM) is greater than the length of the X array (LENX), print a message warning that the X array will have to be enlarged.

7. RETURN.

Flow Chart for Module WEL1AL

MXWELL is the maximum number of wells that will be active at any one time during the simulation.

IWELCB is a flag and a unit number.

If IWELCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set.

If IWELCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IWELCB < 0, well recharge will be printed whenever ICBCFL is set.

LCWELL is a location pointer to the first storage location occupied by the well list.

ISUM is the location of the lowest unallocated storage location in the X array.

X array is the pool of memory space allocated for storing specific tables, arrays, and lists.

LENX is the size of the X array.



```
      SUBROUTINE WEL1AL(ISUM,LENX,LCWELL,MXWELL,NWELLS,IN,IOUT,
     1                  IWELCB)
C
C-----VERSION 1538 12MAY1987 WEL1AL
C     ******************************************************************
C     ALLOCATE ARRAY STORAGE FOR WELL PACKAGE
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
C     ------------------------------------------------------------------
C
C1------IDENTIFY PACKAGE AND INITIALIZE NWELLS
      WRITE(IOUT,1)IN
    1 FORMAT(1H0,'WEL1 -- WELL PACKAGE, VERSION 1, 9/1/87',
     1' INPUT READ FROM',I3)
      NWELLS=0
C
C2------READ MAX NUMBER OF WELLS AND
C2------UNIT OR FLAG FOR CELL-BY-CELL FLOW TERMS.
      READ(IN,2) MXWELL,IWELCB
    2 FORMAT(2I10)
      WRITE(IOUT,3) MXWELL
    3 FORMAT(1H ,'MAXIMUM OF',I5,' WELLS')
      IF(IWELCB.GT.0) WRITE(IOUT,9) IWELCB
    9 FORMAT(1X,'CELL-BY-CELL FLOWS WILL BE RECORDED ON UNIT',I3)
      IF(IWELCB.LT.0) WRITE(IOUT,8)
    8 FORMAT(1X,'CELL-BY-CELL FLOWS WILL BE PRINTED WHEN ICBCFL NOT 0')
C
C3------SET LCWELL EQUAL TO LOCATION OF WELL LIST IN X ARRAY.
      LCWELL=ISUM
C
C4------ADD AMOUNT OF SPACE USED BY WELL LIST TO ISUM.
      ISP=4*MXWELL
      ISUM=ISUM+ISP
C
C5------PRINT NUMBER OF SPACES IN X ARRAY USED BY WELL PACKAGE.
      WRITE(IOUT,4) ISP
    4 FORMAT(1X,I8,' ELEMENTS IN X ARRAY ARE USED FOR WELLS')
      ISUM1=ISUM-1
      WRITE(IOUT,5) ISUM1,LENX
    5 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED OUT OF',I8)
C
C6------IF THERE ISN'T ENOUGH SPACE IN THE X ARRAY THEN PRINT
C6------A WARNING MESSAGE.
      IF(ISUM1.GT.LENX) WRITE(IOUT,6)
    6 FORMAT(1X,'   ***X ARRAY MUST BE DIMENSIONED LARGER***')
C7------RETURN
      RETURN
      END
```

List of Variables for Module WEL1AL

Variable	Range	Definition
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISP	Module	Number of words in the X array allocated by this module.
ISUM	Global	Index number of the lowest element in the X array which has not yet been allocated. When space is allocated for an array, the size of the array is added to ISUM.
ISUM1	Module	ISUM-1.
IWELCB	Package	Flag and a unit number. > 0, unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set. = 0, cell-by-cell flow terms will not be printed or recorded. < 0, well recharge will be printed whenever ICBCFL is set.
LCWELL	Package	Location in the X array of the first element of array WELL.
LENX	Global	Length of the X array in words. This should always be equal to the dimension of X specified in the MAIN program.
MXWELL	Package	Maximum number of wells active at any one time.
NWELLS	Package	Number of wells active during the current stress period.